Supplemental Digital Content 2: NONMEM Control File for Final Labor Progress Model

$PROB Epidural shift, Weight on L1, L2, Asians differ in L1 L2 C

$INPUT ID GRP RTIM DIL=DV MDV

       EFF=DROP EMDV=DROP STA=DROP SMDV=DROP

       NRS NMDV

       TOXY  TEPI  TROM  TIND

       OXY EPI ROM IND

       AGE AMDV=DROP WT WMDV=DROP HT HMDV=DROP GA GMDV=DROP BWT BMDV=DROP

$DATA Uber.Labor.Progress.csv IGNORE=@

      IGNORE=(WT.LT.0)

;Medians (from R)

;Median Age:  29

;Median Weight: 74.2

;Median Height: 163

;Median Gestational Age: 39

;Median Baby Weight: 3.26

$PRED

   DA = AGE - 29

   DW = WT - 74.2

   DH = HT - 163

   DG = GA - 39

   DB = BWT - 3.26

   ; Nominal values

   NL1 = THETA(1)

   NL2 = THETA(2)

   NC1 = THETA(3)

   ;Nominal values for differing ethnicity

   IF (GRP.EQ.1) THEN

      NL1 = THETA(10)

      NL2 = THETA(11)

      NC1 = THETA(12)

   ENDIF

   ;Covariate Effects

   TVL1 = NL1 * EXP(THETA(7)*DW)

   TVL2 = NL2 * EXP(THETA(8)*DW)

   TVC1 = NC1 * EXP(THETA(9)*DW)

   ;Interinividual Variability

   L1 = TVL1 * EXP(ETA(1))

   L2 = TVL2 * EXP(ETA(2))

   C1 = TVC1 * EXP(ETA(3))

   C2 = 10 - C1

   FEPI = THETA(4) ; Epidural Time Factor

   FOXY = THETA(5) ; Oxytocin Time Factor

   FIND = THETA(6) ; Induction Time Factor

   ; Epidural Time Adjustment

   IF(RTIM.LE.TEPI) THEN

      AEPI = RTIM*(1-FEPI)

      REPI = FEPI

   ELSE

      AEPI = TEPI*(1-FEPI)

      REPI = 1

   ENDIF

   ;Oxytocin Time Adjustment

   IF(RTIM.LE.TOXY) THEN

      AOXY = RTIM*(1-FOXY)

      ROXY = FOXY

   ELSE

      AOXY = TOXY*(1-FOXY)

      ROXY = 1

   ENDIF

   ;Induction Time Adjustment

   IF(RTIM.LE.TIND) THEN

      AIND = RTIM*(1-FIND)

      RIND = FIND

   ELSE

      AIND = TOXY*(1-FIND)

      RIND = 1

   ENDIF

   TIME = RTIM-AEPI-AOXY-AIND

   IF(TIME.LT.0) THEN

      EXIT 1

   ENDIF

   IPRD = C1*EXP(-L1*TIME)+C2*EXP(-L2*TIME)

   RATE = (L1*C1*EXP(-L1*TIME)+L2*C2*EXP(-L2*TIME))*REPI*ROXY*RIND

   Y = IPRD + EPS(1)

$THETA

;Nominal parameter values

(.001, 0.56, 3)  ; L1

(.001, 0.09, 1)  ; L2

(0,3.5,10)       ; L3

;Time Scale Factors (0.2 means that if running, labor runs at 20% of the usual rate)

0.6    ; Epidural Time Scale Factor

1 FIX  ; Oxytocin Time Scale Factor

1 FIX  ; Induction Time Scale Factor

;Covariate 1 Effect (weight)

0.0094  ; L1

-0.0099 ; L2

0 FIX   ; L3

;Ethnicity Effect (difference in Asians)

(.001, 0.56, 3)  ; L1

(.001, 0.09, 1)  ; L2

(0,3.5,10)       ; L3

$OMEGA

0.6  ; L1

0.2  ; L2

0.2  ; C1

$SIGMA

2

$EST MAX = 1000 PRINT = 1 NOABORT METHOD = 1 

$TABLE ID RTIM IPRD MDV GRP L1 L2 C1 NOPRINT FILE=AllRecords.txt

$TABLE ID ETA1 ETA2 ETA3 GRP L1 L2 C1 NOPRINT FILE=FirstRecords.txt FIRSTONLY

$TABLE 

      ID RTIM IPRD RATE GRP OXY EPI ROM IND NRS NMDV  

      NOAPPEND

      ONEHEADER

      NOPRINT FILE=allpain.data.txt

